
Otimização com restrições, usando Multiplicadores de 
Lagrange 
Determine o ponto da superficie xy+yz+zx=12 que está mais próximo da origem.

> restart;

> with(plots):

A função objetiva é 

> f:=(x,y,z)->sqrt(x^2+y^2+z^2);

Introduzimos uma função g de tres variaveis tais que a superficie é a superficie de nivel 

g(x,y,z)=0. 

> g:=(x,y,z)->x*y+y*z+z*x-12;

> implicitplot3d(g(x,y,z)=0,x=-6..6,y=-6..6,z=-6..6,axes=boxed,orientation=[-25,76]);



Neste caso podemos tambem realizar esta superficie como grafico da uma função 

z=h(x,y). 

> zz:=solve(g\(x,y,z)=0,z);

> h:=unapply(zz,x,y);

Existe uma descontinuidade obvia ao longo da reta y+x=0 -- no desenho pode aparecer coisas 
estranhas, para obter alguma sensibilidade nós precisamos truncar a variação de z. 
ROTACIONE O seguinte plot.

> plot3d(h,-6..6,-6..6,view=-6..6);



> 

Restrigindo o dominio apenas na parte positiva de x e y obtemos um desenho melhor.

> plot3d(h,0..6,0..6,view=0..6);

> 



O seguinte é o plot mais sofisticado, de fato usamos a função objetiva distância para colorir o 
plot.

Juntamos os eixos coordenados para melhor visualização.

> constr:=a->display([plot3d(h(x,y),x=-a..a,y=-a..a,view=-a..a,axes=boxed,

> color=-sqrt(3/f(x,y,h(x,y)))),

> plot3d({[u,v,0],[u,0,v],[0,u,v]},u=-a..a,v=-a..a,grid=[30,30],style=patchnogrid)],

> orientation=[-20,80]):

> constr(6);

> display(constr(12),orientation=[40,80]);



Os gráficos ilustram que existem dois minimos localizados em alguns pontos 

(r,r,r) e (-r,-r,-r) (usando a simetria da fórmula).

> 

Uma boa maneira de ilustrar o problema de minimizar distancias eh plotar juntos os conjuntos 
de nivel da função objetiva f que são esferas na origem. em vez de perguntar sobre o ponto de 
distância minima sobre a superficie

g(x,y,z)=0, perguntamos pela maior esfera que não intercepta a superficie.

> level:=(r,cc)->sphereplot(r,theta=0..2*Pi,phi=0..Pi,color=cc);

> constr2:=a->plot3d(h(x,y),x=-a..a,y=-a..a,view=-a..a,axes=boxed,

> color=-sqrt(3/f(x,y,h(x,y))), orientation=[-20,80]);

> largest:=r->display({level(r,red),constr2(12)},orientation=[-45,90]);



> largest(2);

> largest(4);

> display(largest(4),orientation=[27,47]);



> largest(2*sqrt(3));

Voltando ao cálculo I : Usamos a restrição para resolver para uma variavel como função das 
outras, então substituir na objetiva f e obter um novo problema de otimização.

Vamos aqui zerar a memoria do MAPLE.



> restart;

> with(plots):

> f:=(x,y,z)->sqrt(x^2+y^2+z^2);

> g:=(x,y,z)->x*y+y*z+z*x-12;

> h:=unapply(solve(g(x,y,z)=0,z),x,y);

> F:=(x,y)->f(x,y,h(x,y));

> F(x,y);

O grafico de F novamente revela a singularidade sobre a reta x+y=0.

Apos alguma tentativa obtemos o seguinte :

(Rotacione o view para identificar os pontos criticos):

> plot3d(F,-6..6,-6..6,view=0..8,style=patchcontour,shading=ZHUE,axes=boxed);



> plot3d(F,-6..6,-6..6,view=0..8,style=patchcontour,shading=ZHUE,

> orientation=[90,0],axes=boxed);

Como no calculo I, para encontrar os extremos de F, nós trabalhamos com (esta poderia 

ter mais pontos criticos, mas não menos!). Vamos determinar o gradiente e igular a zero e 



encontrar os pontos criticos. Vamos usar o teste da derivada segunda para estudar a natureza 
dos pontos criticos.

> FF:=F(x,y)^2;

> FFx:=diff(FF,x);

> FFy:=diff(FF,y);

> sys:=({FFx=0,FFy=0});

> solve(sys,{x,y});

Surpresa o MAPLE determinou as raizes e algumas são complexas.

Em geral esperamos ter que usar o comando fsolve em problemas não lineares.

> fsolve(sys,{x,y},x=-5..5,y=-5..5);

Obtemos apenas soluções negativas -- do nosso estudo sobre o grafico sabemos que existe uma 
raiz tambem no primeiro quadrante; vamos procurar raizes nesta região:

> fsolve(sys,{x,y},x=0..5,y=0..5);

Para completar:



> FFxx:=diff(FFx,x);

> FFxy:=diff(FFx,y);

> FFyy:=diff(FFy,y);

> discr:=FFxx*FFyy-FFxy^2;

> subs({x=2,y=2},discr);

> subs({x=-2,y=-2},discr);

> subs({x=2,y=2},FFxx);

> subs({x=-2,y=-2},FFxx);

O teste da derivada segunda mostra que estes são de fato minimos de FF= , e assim 

tambem minimos de F (note que maximo de FF pode ser minimo ou maximo de F).

> 



Finalmente o enfoque com multiplicadores de Lagrange:

> restart;

> with(plots):

> f:=(x,y,z)->x^2+y^2+z^2;

> g:=(x,y,z)->x*y+y*z+z*x-12;

Vamos novamente maximizar o quadrado da distancia.

Usualmente tomamos o Lagrangiano e igualamos a zero todas as derivadas 
parciais de L de ordem 1. Isto produz um sistema grad(f) = -lambda*grad(g) combinado com a 
restrição g=0.

> L:=f(x,y,z)+lambda*g(x,y,z);

> Lx:=diff(L,x);

> Ly:=diff(L,y);

> Lz:=diff(L,z);

> LL:=diff(L,lambda);

> solve({Lx=0,Ly=0,Lz=0,LL=0},{x,y,z,lambda});

> 

Novamente esperamos dois pontos criticos em (x,y,z)=(2,2,2) e (-2,-2,-2), mais duas soluções que 
após investigação verifica-se complexa.



NOTA: o metodo de multiplicadores de Lagrange apenas dá os pontos criticos -- não discutimos 
um teste de derivada segunda. Naturalmente, para distinguir os pontos criticos bons dos ruins é 
recomendavel, por exemplo, substituir na funçao.


